
The Portal of Tradition 
trad 

An IFF portal is point of entrance; it is a namespace that serves as an interface and is used for communica-
tion between an external representation and the IFF Model Theory Ontology. 

The namespace of models uses terminology that is appropriate for a model theory based upon the theory of 
classifications. The resulting terminology in that namespace is somewhat abstract. The portal of tradition 
uses more elementary terminology than the namespace of models, and hopefully uses terminology that  is 
closer to that of traditional model theory. One can view this traditional portal as a generic portal into the 
IFF Model Theory Ontology. Other portals, such as the portal of conceptual graphs, are more specifically 
oriented to a particular representation. 

Table 1 lists the components of 1st order logic, both syntax and semantics. Non-constant function symbols 
are ignored in this table. 

Table 1: Components of First Order Logic 

LOGIC (SYNTAX) MODEL THEORY (FORMAL SEMANTICS) 
1st order language L = 〈C, P〉  model for language L M = 〈D, I〉  
  domain, universe dom(M) = D 
set of constants const(L) = C interpretation I : const(L) → dom(M) 
set of predicates pred(L) = P  I : pred(L) → ℘ tuple(dom(M)) 
valence valL : pred(L) → Natno   
set of formulas fmla(L) extended interpretation I : fmla(L) → ℘ tuple(dom(M)) 
set of sentences sent(L) ⊆  fmla(L)  I : sent(L) → truth = ℘ 1  ≅   { f, t}  
variable x ∈  var   
constant c ∈  const(L) element, member I[c] ∈  dom(M) 
predicate p ∈  pred(L), valL(p) = n predicate, relation I[p] ⊆  dom(M)n 
formula  φ ∈  fmla(L) predicate, relation I[φ] ⊆  dom(M)n 
sentence  σ ∈  sent(L) truth value I[σ] ∈  truth 
theory Σ ⊆  sent(L)   
entailment  Σ ⊢ σ   
  satisfaction  M ⊨ σ 
theory of a model th(M)   
models of a theory mod(Σ)   

Any model A in the IFF Model Theory Ontology has an associated traditional sorted model of the type 
language L = typ(A). The following comments compare the components of a model A in the IFF Model 
Theory Ontology with both the interpretations in Table 1 and standard logic definitions (see Enderton) for 
many-sorted languages and logics.  

http://grouper.ieee.org/groups/suo/email/pdf00003.pdf
http://grouper.ieee.org/groups/suo/email/msg01792.html

Table 2 lists the terminology in the portal of tradition. 

Table 2: The terminology of the tradition portal 

 Class Function 

trad 
.lang 

traditional-language variable entity reference 
relation arity signature 
language 

  expression sentence 
atom negation 
conjunction disjunction implication equivalence 
existential-quantification universal-quantification 
substitution 

trad 
.mod 

traditional-model variable universe pair entity-type type-reference  
entity-extent entity 
reference 
tuple relation-type type-arity type-signature 
relation-extent relation 
signature 
model 
type 

  satisfaction 

 

Logic (Syntax) 
trad.lang 

The notion of a traditional 1st-order language in Table 1 corresponds to an IFF type language L. There are 
two differences: the IFF type languages are sorted, whereas the languages described in Table 1 are un-
sorted; the languages in the current version of the IFF Model Theory Ontology do not have constants, func-
tion symbols or terms – the next version of the IFF Model Theory Ontology will incorporate these. Tradi-
tional languages will be mapped to IFF type languages below. 
○ A traditional (sorted) 1st-order language L has the following components. 

− a set of variables var(L) 
− a set of entity types ent(L) 
− a reference (sort) function ∗ L = refer(L) : var(L) → ent(L) 
Enderton’s sort set S is represented by the set of entity types ent(L) = tgt(refer(L)) and his partition of 
sorted variables { vars | s ∈  S}  is represented by the reference function refer(L). The set of variables 
var, either in Table 1 or in Enderton, corresponds to the set of variables var(L) = src(refer(L)).  
Note, however that each language contains its own set of variables. Morphisms between languages 
have a component function between their respective variable sets. Although such variable functions are 
unconstrained for Language, the category of IFF type languages, they are constrained to be bijections 
for IFF model morphisms. We can have variable sets of any cardinality. However, two models are only 
connected by a morphism, if their variable sets have the same cardinality.  
The set of variables var(L) generalizes the usual case where natural numbers and sequences are used – 
the advantage for this generality is elimination of the dependency on sequences and natural numbers.  
Sequence, along with other traditional data structures, will be represented by the IFF Data Structures 
Ontology in the lower metalevel. One way to return to sequence-like structures is to define a set of in-
dexed variables var(L) = index×ent(L) for some index set such as index = Natno, and to let the refer-
ence function be defined as the projection refer(L) = π : index×ent(L) → ent(L). 
(1) (SET$class traditional-language) 
 
(2) (SET.FTN$function variable) 
    (= (SET.FTN$source variable) traditional-language) 
    (= (SET.FTN$target variable) set$set) 
 
(3) (SET.FTN$function entity) 

    (= (SET.FTN$source entity) traditional-language) 
    (= (SET.FTN$target entity) set$set) 
 
(4) (SET.FTN$function reference) 
    (= (SET.FTN$source reference) traditional-language) 
    (= (SET.FTN$target reference) set.ftn$function) 
    (= (SET.FTN$composition [reference set.ftn$source]) variable) 
    (= (SET.FTN$composition [reference set.ftn$target]) entity) 

○ Continuing, a traditional (sorted) 1st-order language L has the following additional components. 
− a set of relation types rel(L) 
− an arity function #A = arity(L) : rel(L) → ℘ var(L), or equivalently, 
− a signature function ∂A = sign(L) : rel(L) → sign(refer(L)) 
where the signature of a relation type element ρ ∈  rel(L) is the restriction of the reference function to 
its arity(L)(ρ):  

sign(L)(ρ)(x) = refer(L)(x) for all variables x ∈  arity(L)(ρ). 
The set of predicates in Table 1 corresponds to the set of relation types rel(L) = src(sign(L)). The set 
of variables arity(L)(ρ) that reference the entity types in the signature of a relation type ρ ∈  rel(L) is 
called its arity. The arity of relation types can overlap. The number |arity(L)(ρ)| of variables in the arity 
of a relation type is called its valence. This corresponds to the valence in Table 1. 
Enderton’s (possibly empty) set of predicates of sort w for each tuple w of sorts corresponds to the 
(possibly empty) set sign(L)−1(τ) ⊆  rel(L) of relation types of signature τ for each signature of entity 
types τ ∈  sign(refer(L)). Such a signature is a function τ : arity(refer(L))(τ) ⊆  var(L) → ent(L), 
which is a restriction of the reference function: τ(x) = refer(L)(x) for all x ∈  arity(refer(L))(τ). Thus, 
the sort-indexed set of predicates corresponds to the signature function sign(L). This function is 
equivalent to the arity function arity(L). 
(5) (SET.FTN$function relation) 
    (= (SET.FTN$source relation) traditional-language) 
    (= (SET.FTN$target relation) set$set) 
 
(6) (SET.FTN$function arity) 
    (= (SET.FTN$source arity) traditional-language) 
    (= (SET.FTN$target arity) set.ftn$function) 
    (= (SET.FTN$composition [arity set.ftn$source]) relation) 
    (= (SET.FTN$composition [arity set.ftn$target]) 
       (SET.FTN$composition [variable set$power])) 
 
(7) (SET.FTN$function signature) 
    (= (SET.FTN$source signature) traditional-language) 
    (= (SET.FTN$target signature) set.ftn$function) 
    (= (SET.FTN$composition [signature set.ftn$source]) relation) 
    (= (SET.FTN$composition [signature set.ftn$target]) 
       (SET.FTN$composition [reference set.ftn$signature])) 
 
(8) (forall (?l (traditional-language ?l) 
             ?rho ((relation ?l) ?rho)) 
        (and (= (set.ftn$composition [((signature ?l) ?rho) set.ftn$source])  
                ((arity ?l) ?rho)) 
             (= (set.ftn$composition [((signature ?l) ?rho) set.ftn$target])  
                (entity ?l)) 
             (set.ftn$restriction ((signature ?l) ?rho) (reference ?l)))) 

○ Any traditional language L is an IFF type language lang(L). This term marshals all the elements of a 
traditional language and maps into the type language namespace of the IFF Model Theory Ontology.  
(9) (SET.FTN$function language) 
    (= (SET.FTN$source language) traditional-language) 
    (= (SET.FTN$target language) lang$language) 
    (= (SET.FTN$composition [language lang$reference]) reference) 
    (= (SET.FTN$composition [language lang$signature]) signature) 

○ For convenience of reference and access, additional terminology of a traditional language is given: 

− the set of expressions expr(L),  
− the various kinds of expressions 

!" atomic expression atom(L),  
!" negations neg(L),  
!" etcetra, and 

− the expressions of empty arity called sentences sent(L).  
Formulas are called expressions in the IFF Model Theory Ontology. The set of formulas fmla(L) in 
Table 1 corresponds to the set of expressions expr(L) = rel(expr(lang(L))), the relation types of the 
expression type language expr(lang(L)) that extends the type language lang(L) from relation types to 
expressions, but shares the same set of variables and the same set of entity types. To reiterate, in the 
IFF Model Theory Ontology expressions are relations types. Sentences are defined to be the set of ex-
pressions having empty arity sent(L) = arity(expr(lang(L)))−1(∅ ) ⊆  rel(expr(lang(L))). This is the ∅ -
fiber of the expression arity function 

arity(expr(lang(L))) : rel(expr(lang(L))) → ℘ var(L). 
(10) (SET.FTN$function expression) 
     (= (SET.FTN$source expression) traditional-language) 
     (= (SET.FTN$target expression) set$set) 
     (= expression (SET.FTN$composition [language lang.expr$set])) 
 
(11) (SET.FTN$function atom) 
     (= (SET.FTN$source atom) traditional-language) 
     (= (SET.FTN$target atom) set$set) 
     (= atom (SET.FTN$composition [language lang.expr$atom])) 
 
(12) (SET.FTN$function negation) 
     (= (SET.FTN$source negation) traditional-language) 
     (= (SET.FTN$target negation) set$set) 
     (= negation (SET.FTN$composition [language lang.expr$negation])) 
 
(13) (SET.FTN$function conjunction) 
     (= (SET.FTN$source conjunction) traditional-language) 
     (= (SET.FTN$target conjunction) set$set) 
     (= conjunction (SET.FTN$composition [language lang.expr$conjunction])) 
 
(14) (SET.FTN$function disjunction) 
     (= (SET.FTN$source disjunction) traditional-language) 
     (= (SET.FTN$target disjunction) set$set) 
     (= disjunction (SET.FTN$composition [language lang.expr$disjunction])) 
 
(15) (SET.FTN$function implication) 
     (= (SET.FTN$source implication) traditional-language) 
     (= (SET.FTN$target implication) set$set) 
     (= implication (SET.FTN$composition [language lang.expr$implication])) 
 
(16) (SET.FTN$function equivalence) 
     (= (SET.FTN$source equivalence) traditional-language) 
     (= (SET.FTN$target equivalence) set$set) 
     (= equivalence (SET.FTN$composition [language lang.expr$equivalence])) 
 
(17) (SET.FTN$function existential-quantification) 
     (= (SET.FTN$source existential-quantification) traditional-language) 
     (= (SET.FTN$target existential-quantification) set$set) 
     (= existential-quantification  
        (SET.FTN$composition [language lang.expr$existential-quantification])) 
  
(18) (SET.FTN$function universal-quantification) 
     (= (SET.FTN$source universal-quantification) traditional-language) 
     (= (SET.FTN$target universal-quantification) set$set) 
     (= universal-quantification  
        (SET.FTN$composition [language lang.expr$universal-quantification])) 
 
(19) (SET.FTN$function substitution) 
     (= (SET.FTN$source substitution) traditional-language) 

     (= (SET.FTN$target substitution) set$set) 
     (= substitution (SET.FTN$composition [language lang.expr$substitution])) 
 
(20) (SET.FTN$function sentence) 
     (= (SET.FTN$source sentence) traditional-language) 
     (= (SET.FTN$target sentence) set$set) 
     (= sentence (SET.FTN$composition [language lang.expr$sentence])) 

Model Theory (Formal Semantics) 
trad.mod 

The extent infomorphisms (Figure 1) of the entity, relation and expression classifications reveal the various 
components of the interpretation functions of a traditional model A. 

 

○ A traditional (sorted) 1st-order model A has the following components. 
− a set of variables var(A) 
− a universe of discourse univ(A) 
− a pair of sets pr(A) = 〈var(A)), univ(A)〉  
− a set of entity types ent-typ(A) 
− a reference (sort) function ∗ A = typ-refer(A) : var(A) → ent-typ(A) 
− an entity extent function ent-ext(A) : ent-typ(A) →℘ univ(A), or equivalently, 

− an entity classification ent(A) = 〈univ(A), ent-typ(A), ⊨ent(A)〉 , and 
− a reference semidesignation refer(A) = 〈pr(A), typ-refer(A)〉  : var(A) ↬ ent(A) (Figure 2a). 

The domain or universe dom(M) = D in Table 1 corresponds to univ(A) = inst(ent(A)), the universe 
(or universal domain) for A, which is the set of instances of the entity classification ent(A) = 
inst(ent(A)). Enderton’s set of sorts S corresponds to the set of entity types ent-typ(A), which is the 
set of types of the entity classification ent-typ(A) = typ(ent(A)). Enderton’s interpretation assignment 
of a set (intuitively, a type) Ts to each sort s ∈  S is represented by the entity extent function ent-ext(A), 
which is the extent function of the entity classification  

ent-ext(A) = extent(A) :  typ(ent(A)) →℘ univ(A).  
This is the type function (Figure 1, right side) of the canonical eta infomorphism 

ηent(A) = 〈 id, extent(A)〉  : ent(A) ⇄℘ univ(A).  
But set-valued functions and classifications are equivalent, with each set-valued function inducing, by 
fiber or inverse image, a classification, and each classification having a set-valued extent function from 
its set of types to the power set of its set of instances. The induced classification of the set-valued en-

Figure 1: Entity, Relation and Expression Extent Infomorphisms 

typ(ent(A)) 

univ(A) 

⊨rel(A) 

℘ univ(A) 

∈ univ(A) 

extent(A) 
typ(rel(A)) 

⊨rel(A) 

extrel(A) 

typ(expr(A)) 

tuple(A) 

⊨expr(A) 

℘ tuple(A) 

∈ tuple(A) 
extexpr(A) 

injrel(A) 

Figure 2a: Reference Semidesignation 

Figure 2b: Signature Designation 

tity extent function is the entity classification induc(ent-ext(A)) = ent(A). Both the entity classifica-
tion and the universe instance power  

inst-pow(univ(A)) = 〈univ(A), ℘ univ(A), ∈ univ(A)〉  
have universe elements as their instances. The following axiomatization builds the reference 
semidesignation of Figure 2a. 
(1) (SET$class traditional-model) 
 
(2) (SET.FTN$function variable) 
    (= (SET.FTN$source variable) traditional-model) 
    (= (SET.FTN$target variable) set$set) 
 
(3) (SET.FTN$function universe) 
    (= (SET.FTN$source universe) traditional-model) 
    (= (SET.FTN$target universe) set$set) 
 
(4) (SET.FTN$function pair) 
    (= (SET.FTN$source pair) traditional-model) 
    (= (SET.FTN$target pair) set.pr$pair) 
    (= (SET.FTN$composition [pair set.pr$set1]) variable) 
    (= (SET.FTN$composition [pair set.pr$set2]) universe) 
 
(5) (SET.FTN$function entity-type) 
    (= (SET.FTN$source entity-type) traditional-model) 
    (= (SET.FTN$target entity-type) set$set) 
 
(6) (SET.FTN$function type-reference) 
    (= (SET.FTN$source type-reference) traditional-model) 
    (= (SET.FTN$target type-reference) set.ftn$function) 
    (= (SET.FTN$composition [type-reference set.ftn$source]) variable) 
    (= (SET.FTN$composition [type-reference set.ftn$target]) entity-type) 
 
(7) (SET.FTN$function entity-extent) 
    (= (SET.FTN$source entity-extent) traditional-model) 
    (= (SET.FTN$target entity-extent) set.ftn$function) 
    (= (SET.FTN$composition [entity-extent set.ftn$source]) entity-type) 
    (= (SET.FTN$composition [entity-extent set.ftn$target]) 
       (SET.FTN$composition [universe set$power])) 
 
(8) (SET.FTN$function entity) 
    (= (SET.FTN$source entity) traditional-model) 
    (= (SET.FTN$target entity) cls$classification) 
    (= (SET.FTN$composition [entity cls$instance]) universe) 
    (= (SET.FTN$composition [entity cls$type]) entity-type) 
    (= entity (SET.FTN$composition [entity-extent set.ftn$induction])) 
    (= entity-extent (SET.FTN$composition [entity cls$extent])) 
 
(9) (SET.FTN$function reference) 
    (= (SET.FTN$source reference) traditional-model) 
    (= (SET.FTN$target reference) cls.sdsgn$semidesignation) 
    (= (SET.FTN$composition [reference cls.sdsgn$set]) variable) 
    (= (SET.FTN$composition [reference cls.sdsgn$classification]) entity) 
    (= (SET.FTN$composition [reference cls.sdsgn$instance]) pair) 
    (= (SET.FTN$composition [reference cls.sdsgn$type]) type-reference) 

○ Continuing, a traditional (sorted) 1st-order model A has the following additional components.  
− a set of relation tuples tuple(A) = tuple(refer(A)) 
− a set of relation types rel-typ(A) 
− an arity function #A = typ-arity(A) : rel-typ(A) → ℘ var(A), or equivalently, 
− a signature function ∂A = typ-sign(A) : rel-typ(A) → sign(typ-refer(A)) 
− a relation extent function rel-ext(A) : rel-typ(A) →℘ tuple(A), or equivalently,  
− a relation classification rel(A) = 〈 tuple(A), rel-typ(A), ⊨rel(A)〉 , and 
− a signature designation sign(A) = 〈 idtuple(A), typ-sign(A)〉  : rel(A) ⇉ sign(refer(A))  (Figure 2b). 

The set tuple(dom(M)) in Table 1 corresponds to tuple(A), the set of tuples or (relational) arguments 
of (the reference semidesignation of) A, which is the set of instances of the relation classification tu-
ple(A) = inst(rel(A)). The set of predicates pred(L) in Table 1, and Enderton’s sort-indexed set of 
predicates, correspond to the set of relation types rel-typ(A), which is the set of types of the relation 
classification rel-typ(A) = typ(rel(A)). 
The predicate interpretation map I : pred(L) → ℘ tuple(dom(M)) in Table 1, and Enderton’s interpre-
tation assignment of a subset of Ts1× … ×Tsn to each predicate of sort (s1, …, sn), are represented by 
the composition of the relation extent type function 

rel-ext(A) : typ(rel(A)) →℘ tuple(A) 
with the power of the instance signature function  

℘∂ 0 : ℘ tuple(A) →℘ tuple(refer(A)) 
and the power of the tuple inclusion function 

℘ incl : ℘ tuple(refer(A)) →℘ tuple(univ(A)). 
The relation extent function is the extent function of the relation classification  

rel-ext(A) = extrel(A) : typ(rel(A)) →℘ tuple(A). 
This is the type function (Figure 1, left side) of the canonical eta infomorphism 

ηrel(A) = 〈 id, extrelA)〉  : rel(A) ⇄℘ tuple(A).  
Using again the equivalency between set-valued functions and classifications, the induced classifica-
tion of the set-valued relation extent function is the relation classification induc(rel-ext(A)) = rel(A). 
Both the relation classification and the tuple instance power  

inst-pow(tuple(A)) = 〈 tuple(A), ℘ tuple(A), ∈ tuple(A)〉  
have tuples as instances.  
The relation extent of any relation type ρ ∈  rel-typ(A) is constrained to tuples r ∈  tuple(A) that are 
more specialized than ρ:  

r ∈  rel-ext(A)(ρ) implies arity(pr(A))(r) ⊇  typ-arity(A)(ρ).  
The relation extent function respects arities (or signatures) in the sense that for any relation type 
ρ ∈  rel-typ(A),  

r ∈  rel-ext(A)(ρ) implies r(x) ∈  ent-ext(A)(typ-refer(A)(x)) for all x ∈  typ-arity(A)(ρ). 
The latter constraint can be expressed in terms of the relation and entity classifications 

r ⊨rel(A) ρ implies r(x) ⊨ent(A) ∂A(ρ)(x) for all x ∈  typ-arity(A)(ρ). 
This is the constraint required of the signature designation. The following axiomatization builds the 
signature designation of Figure 2b. 
(10) (SET.FTN$function tuple) 
     (= (SET.FTN$source tuple) traditional-model) 
     (= (SET.FTN$target tuple) set$set) 
     (= tuple (SET.FTN$composition [reference cls.sdsgn$tuple])) 
 
(11) (SET.FTN$function relation-type) 
     (= (SET.FTN$source relation-type) traditional-model) 
     (= (SET.FTN$target relation-type) set$set) 
 
(12) (SET.FTN$function type-arity) 
     (= (SET.FTN$source type-arity) traditional-model) 
     (= (SET.FTN$target type-arity) set.ftn$function) 
     (= (SET.FTN$composition [type-arity set.ftn$source]) relation-type) 
     (= (SET.FTN$composition [type-arity set.ftn$target]) 
        (SET.FTN$composition [variable set$power])) 
 
(13) (SET.FTN$function type-signature) 
     (= (SET.FTN$source type-signature) traditional-model) 
     (= (SET.FTN$target type-signature) set.ftn$function) 
     (= (SET.FTN$composition [type-signature set.ftn$source]) relation-type) 
     (= (SET.FTN$composition [type-signature set.ftn$target]) 

        (SET.FTN$composition [type-reference set.ftn$signature])) 
 
(14) (SET.FTN$function relation-extent) 
     (= (SET.FTN$source relation-extent) traditional-model) 
     (= (SET.FTN$target relation-extent) set.ftn$function) 
     (= (SET.FTN$composition [relation-extent set.ftn$source]) relation-type) 
     (= (SET.FTN$composition [relation-extent set.ftn$target]) 
        (SET.FTN$composition [tuple set$power])) 
 
(15) (forall (?a (traditional-model ?a) ?rho ((relation-type ?a) ?rho) ?r ((tuple ?a) ?r)) 
         (=> (((relation-extent ?a) ?rho) ?r) 
             (and (set$subset  
                      ((type-arity ?a) ?rho) ((cls.sdsgn$tuple-arity (reference ?a)) ?r)) 
                  (forall (?x (((type-arity ?a) ?rho) ?x)) 
                      (((entity-extent ?a) ((type-reference ?a) ?x)) (?r ?x)))))) 
 
(16) (SET.FTN$function relation) 
     (= (SET.FTN$source relation) traditional-model) 
     (= (SET.FTN$target relation) cls$classification) 
     (= (SET.FTN$composition [relation cls$instance]) tuple) 
     (= (SET.FTN$composition [relation cls$type]) relation-type) 
     (= relation (SET.FTN$composition [relation-extent set.ftn$induction])) 
     (= relation-extent (SET.FTN$composition [relation cls$extent])) 
 
(17) (forall (?a (traditional-model ?a) ?rho ((relation-type ?a) ?rho) ?r ((tuple ?a) ?r)) 
         (=> ((relation ?a) ?r ?rho) 
             (and (set$subset  
                      ((type-arity ?a) ?rho) ((cls.sdsgn$tuple-arity (reference ?a)) ?r)) 
                  (forall (?x (((type-arity ?a) ?rho) ?x)) 
                      ((entity ?a) (?r ?x) ((type-reference ?a) ?x)))))) 
 
(18) (SET.FTN$function signature) 
     (= (SET.FTN$source signature) traditional-model) 
     (= (SET.FTN$target signature) cls.dsgn$designation) 
     (= (SET.FTN$composition [signature cls.dsgn$source]) relation) 
     (= (SET.FTN$composition [signature cls.dsgn$target]) 
        (SET.FTN$composition [reference cls.sdsgn$signature])) 
     (= (SET.FTN$composition [signature cls.dsgn$instance])  
        (SET.FTN$composition [tuple set.ftn$identity])) 
     (= (SET.FTN$composition [signature cls.dsgn$type]) type-signature) 

○ Any traditional model A is an IFF model mod(A). This term marshals all the elements of a traditional 
model and maps into the model namespace of the IFF Model Theory Ontology. 
(19) (SET.FTN$function model) 
     (= (SET.FTN$source model) traditional-model) 
     (= (SET.FTN$target model) mod$model) 
     (= (SET.FTN$composition [model mod$reference]) reference) 
     (= (SET.FTN$composition [model mod$signature]) signature) 

○ The type aspect of any traditional model A is a traditional language L. The IFF type language of the 
type aspect of any traditional model is the type language of the its IFF model. 
(20) (SET.FTN$function type) 
     (= (SET.FTN$source type) traditional-model) 
     (= (SET.FTN$target type) traditional-language) 
     (= (SET.FTN$composition [type trad.lang$variable]) variable) 
     (= (SET.FTN$composition [type trad.lang$entity]) entity-type) 
     (= (SET.FTN$composition [type trad.lang$reference]) type-reference) 
     (= (SET.FTN$composition [type trad.lang$relation]) relation-type) 
     (= (SET.FTN$composition [type trad.lang$arity]) type-arity) 
     (= (SET.FTN$composition [type trad.lang$signature]) type-signature) 
 
(21) (= (SET.FTN$composition [type trad.lang$language]) 
        (SET.FTN$composition [model mod$type])) 

○ For convenience of reference and access, additional terminology of a traditional model is given: 
− an expression extent function expr-ext(A) : expr(A) →℘ tuple(A), or equivalently,  
− an expression classification expr-cls(A) = 〈 tuple(A), expr(A), ⊨expr(A)〉 , 

− a satisfaction relation (subset of satisfied sentences) sat(A) ⊆  sent(lang(A)). 
The formula interpretation map I : fmla(L) → ℘ tuple(dom(M)) in Table 1 is represented by the com-
position of the expression extent type function 

extexpr-cls(A) : typ(expr-cls(A)) →℘ tuple(A) 
with the power of the instance signature function  

℘∂ 0 : ℘ tuple(A) →℘ tuple(refer(A)) 
and the power of the tuple inclusion function 

℘ incl : ℘ tuple(refer(A)) →℘ tuple(univ(A)). 
The expression extent function is the extent function of the expression classification  

expr-ext(A) = extexpr-cls(A) : typ(expr-cls(A)) →℘ tuple(A). 
This is the type function (Figure 1, left side) of the canonical eta infomorphism 

ηexpr-cls(A) = 〈 id, extexpr-cls(A)〉  : expr-cls(A) ⇄℘ tuple(A).  
Using again the equivalency between set-valued functions and classifications, the induced classifica-
tion of the set-valued expression extent function is the expression classification induc(expr-ext(A)) = 
expr-cls(A). Both the expression classification and the tuple instance power  

inst-pow(tuple(A)) = 〈 tuple(A), ℘ tuple(A), ∈ tuple(A)〉  
have tuples as instances.  
The satisfaction relation M ⊨ σ in Table 1 between traditional models and sentences is represented by 
the satisfies subset sat(A) ⊆  sent(lang(A)).  
(22) (SET.FTN$function expression-extent) 
     (= (SET.FTN$source expression-extent) traditional-model) 
     (= (SET.FTN$target expression-extent) set.ftn$function) 
     (= (SET.FTN$composition [expression-extent set.ftn$source]) expression) 
     (= (SET.FTN$composition [expression-extent set.ftn$target]) 
       (SET.FTN$composition [(SET.FTN$composition [model mod$universe]) set$power])) 
       (= expression-extent (SET.FTN$composition [model mod$extent])) 
 
(23) (SET.FTN$function expression-classification) 
     (= (SET.FTN$source expression-classification) traditional-model) 
     (= (SET.FTN$target expression-classification) cls$classification) 
     (= (SET.FTN$composition [expression-classification cls$instance]) tuple) 
     (= (SET.FTN$composition [expression-classification cls$type]) 
        (SET.FTN$composition [type trad.lang$expression])) 
     (= expression-classification  
       (SET.FTN$composition [model mod$expression-classification])) 
 
(24) (SET.FTN$function satisfies) 
     (= (SET.FTN$source satisfies) traditional-model) 
     (= (SET.FTN$target satisfies) set$set) 
     (= satisfies (SET.FTN$composition [model mod$satisfies])) 

